Chapter 1

Memristor SPICE Modeling

Chris Yakopcic, Tarek M. Taha, Guru Subramanyam, and Robinson E. Pino

Abstract Modeling of memristor devices is essential for memristor based circuit and system design. This chapter presents a review of existing memristor modeling techniques and provides simulations that compare several existing models to published memristor characterization data. A discussion of existing models is presented that explains how the equations of each relate to physical device behaviors.

The simulations were completed in LTspice and compare the output of the different models to current-voltage relationships of physical devices. Sinusoidal and triangular pulse inputs were used throughout the simulations to test the capabilities of each model. The chapter is concluded by recommending a more generalized memristor model that can be accurately matched to several different published device characterizations. This generalized model provides the potential for more accurate circuit simulation for a wide range of device structures and voltage inputs.

1.1 Introduction

The memristor was theorized in 1971 by Dr. Leon Chua [1], and was first fabricated by a research team led by Dr. Stanley Williams at HP Labs in 2008 [2, 3]. The memristor is a non-volatile nanoscale 2-termial passive circuit element that has dynamic resistance dependent on the total charge applied between the positive and negative terminals.

The memristor fabricated at HP Labs was a thin-film titanium oxide device. The device structure comprised of a stoichiometric (TiO_2) and an oxygen deficient (TiO_{2-x}) layer sandwiched between two platinum electrodes. Applying a voltage across a memristor causes the oxygen deficiencies in the TiO_{2-x} layer to migrate,

C. Yakopcic, T. M. Taha, and G. Subramanyam
Department of Electrical and Computer Engineering
University of Dayton, Dayton, Ohio, USA
e-mail: {yakopccg, tarek.taha, guru.subramanyam}@notes.udayton.edu
R. E. Pino
Information Directorate, Advanced Computing Architectures
Air Force Research Laboratory, Rome, NY, USA
e-mail: robinson.pino@rl.af.mil

and this changes the thickness of the oxygen deficient layer. Likewise, this changes the resistance of the memristor device. Since the oxygen vacancies have a low mobility, they tend to stay in the same position after the voltage source is removed [3]. This phenomenon shows that the memristor can be used as non-volatile memory device where the resistance of a memristor is used to store information.

Applications where memristors may be used include high density non-volatile memory [4] and logic design [5, 6]. One of the more interesting applications for the memristor involves using the device to mimic the functionality of a synapse in brain tissue [7, 8]. Just as neural spikes are applied to a synapse to change the weight, voltage pulses can be applied to a memristor to change the resistance. Since the dynamics of a memristor closely model a synapse [7, 8], memristors are considered ideal for spiking input based neuromorphic systems. Thus the simulation of memristors for spiking inputs is essential to the design of memristor based neuromorphic systems.

Since the initial fabrication and modeling efforts by HP Labs [2], several different memristor device structures and materials have been published [7-12]. The wide variety in memristor structure and composition has led to the development of many different memristor modeling techniques. Several compact models have been proposed that present modeling equations that approximate the functionality of published memristor devices [2, 13–16]. Furthermore, a number of subcircuits have been proposed that provide the capability of modeling memristors in SPICE simulations [17–24]. Many of these models [14, 17, 19-22], are based on the memristor equations first proposed by HP Labs in [2]. Additionally, advances in modeling have been published [25] based on the original memristor equations proposed by Dr. Chua [1]. The remainder of the memristor models are either closely correlated to device hardware [13, 16, 24], and/or based on more complex physical mechanisms [15, 18, 23, 24] such as the metal-insulator-metal (MIM) tunnel junction [26].

This chapter provides a review of many of the different memristor modeling techniques [2, 13-17, 23, 24]. The memristor models chosen to be discussed in this chapter were selected to show a wide variety of different modeling techniques while minimizing redundancy. Some models have been designed to represent a specific device very accurately, and other models aim to reproduce the functionality of a wider range of devices in a more generalized manner.

The model results in this chapter are discussed in terms of their resulting I-V characteristics and how well they model the I-V characteristics of physical devices. The voltage inputs studied are either sinusoidal or triangular pulses. The triangular input pulses are applied multiple times with the same polarity to study how each model switches to intermediate levels between the maximum and minimum resistance. The SPICE simulations were performed in LTspice, and the subcircuit code is provided for each model. This allowed for a one-to-one comparison of many different memristor models to show the advantages and disadvantages of each.

This chapter is organized as follows: Section 1.2 describes the how the memristor modeling equations were developed based on the initial memristor fabrication

at HP Labs. Section 1.3 shows how these initial equations were modified to develop SPICE models. Section 1.4 describes two alternative SPICE modeling techniques where the model output of each correlates very closely to the characterization data of a specific device. Section 1.5 describes models that have been developed based on a hyperbolic sinusoid current-voltage relationship. This appears to improve the model result when using repetitively pulsed inputs. Section 1.6 discusses a generalized SPICE model that can be used to accurately model the current-voltage relationship of several different memristor devices. Section 1.7 provides a conclusion that summarizes the results.

1.2 Memristor Model Proposed by HP Labs

In 2008, HP Labs published results that described the memristor device according to equations (1.1) through (1.3). The current voltage relationship is described in equation (1.1). The value of $I(t)$ represents the current through the memristor, and the voltage $V(t)$ represents the voltage generated at the input source. These definitions for $I(t)$ and $V(t)$ are used consistently throughout this chapter. The constants R_{OFF} and R_{ON} represent the maximum and minimum resistances of the device respectively. The actual resistance of the device is dependent on the ratio between the value of the dynamic state variable $w(t)$ and the device thickness D. The state variable $w(t)$ represents the thickness of the oxygen deficient titanium dioxide layer (TiO_{2-x}). As the value of $w(t)$ increases, it can be seen that the overall device resistance lowers since $R_{OFF} > R_{ON}$.

The dynamic value of the state variable can be determined using equation (1.2) where dw/dt is described as the drift velocity of the oxygen deficiencies (v_D) in the device. The value for $w(t)$ can be determined by integrating equation (1.2), and the result can be seen in equation (1.3). After integration, it can be seen that the value for $w(t)$ is proportional to the charge on the device. Since the charge is the integral of the current, this provides an explanation for the non-volatile effect of the memristors: when no current is flowing through the device, the charge is constant and thus, the resistance remains unchanged.

$$V(t) = \left[R_{ON} \frac{w(t)}{D} + R_{OFF} \left(1 - \frac{w(t)}{D} \right) \right] I(t) \tag{1.1}$$

$$v_D = \frac{dw}{dt} = \frac{\mu_D R_{ON}}{D} I(t) \tag{1.2}$$

$$w(t) = \frac{\mu_D R_{ON}}{D} q(t) \tag{1.3}$$

Fig. 1.1 shows how this memristor model reacts to a simple sinusoidal voltage input. The I-V curve displays a pinched hysteresis loop that is characteristic of memristors. The hysteresis shows that the conductivity in a memristor is not only related to the voltage applied, but also to the previous value of the state variable $w(t)$, as more than one current value can be correlated to a single voltage. Fig. 1.2 shows the simulation results of the model when several triangular voltage pulses are applied to the device. The first 4 voltage pulses, occurring in the time between 0 and 4 seconds, correspond to the right half of the I-V curve. Since the charge applied is always positive, the state variable continually increases. This results in an increase in the conductivity of the device as each pulse in applied. For the simulation time between 4 and 8 seconds, the current is always negative, and the opposite trend can be seen. These simulations were performed in MATLAB assuming that the voltage signal was directly applied to a memristor device with no additional circuit elements or added resistances. The next section discusses how these equations were used to generate a SPICE subcircuit.



Fig. 1.1 Simulation results for the HP Labs memristor with a sinusoidal input. In this simulation: R_{ON}=10kΩ, R_{OFF}=100kΩ, μ_D=10^{-14}m^{2}s^{-1}V^{-1}, D=27nm, x_0=0.1D, and $V(t)$=sin(2πt).



Fig. 1.2 Simulation results for the HP Labs memristor with a pulsed input. In this simulation: R_{ON}=10kΩ, R_{OFF}=100kΩ, μ_D=10^{-14}m^{2}s^{-1}V^{-1}, D=27nm, x_0=0.1D, and $V(t)$=sin(2πt). Triangular pulses have magnitude of 1V and 1 second pulse width with rise and fall time of 0.5 seconds.

1.3 Initial Memristor SPICE Modeling

It is of great benefit to circuit designers to be able to model the memristor in SPICE simulators, therefore equations (1.1) and (1.2) were used to develop SPICE subcircuits for memristors [17-24]. Modifications to these initial equations were made to develop a robust technique for simulating memristors based on these initial equations.

First, it should be noted that the state variable equations were updated using the variable substitution $x(t)=w(t)/D$. The state variable is now a normalized quantity between 0 and 1. When $x(t)=0$ the memristor device is in the least conductive state, and the most conductive state occurs when $x(t)=1$. Memristor devices have been proposed using several different material structures [7-12], so the resistance switching mechanism is not always due to the change in thickness of a titanium oxide layer. This change in state variable represents a generalization of the model so that it can represent more than just titanium oxide devices.

Next, the state variable boundaries were defined. The modeling equations in Section 1.2 do not account for the state variable boundaries: $0 \leq x(t) \leq 1$ (or $0 \leq w(t) \leq D$). If sufficient charge is applied to the memristor, then the value of $x(t)$ will become larger than 1 and the result of the model will become unstable and thus incorrect. The state variable motion was limited by using two different windowing functions [14, 17], and SPICE models were developed based on these modified equations.

1.3.1 Joglekar Modifications

In a publication by Yogesh N. Joglekar and Stephen J. Wolf [14], modifications were made to the initial equations proposed by HP Labs [2]. The parameter η was added so that memristors could be modeled where state variable motion could be in either direction relative to the input voltage. If a positive voltage signal applied to a memristor increases the value of the state variable, then the memristor device should be modeled where $\eta=1$. If the state variable decreases with the application of a positive voltage signal, then that memristor should be modeled with $\eta=-1$.

Additionally, the windowing function in equation (1.4) was added to the equation for state variable motion. This was done to ensure that the state variable will always fall in the range $0 \leq x(t) \leq 1$. Fig. 1.3 displays plots for the Joglekar and Biolek (see Section 1.3.2) window functions for all acceptable values of $x(t)$. When looking at the left plot in Fig. 1.3, it can be seen that the Joglekar window function forces the state variable motion to be zero at $x(t)=0$ or $x(t)=1$, thus defining the boundaries. Depending on the value of the parameter p, the window function can provide a harder boundary effect (where $p=100$), or provide a smoother non-linearity in the motion of the state variable (where $p=1$). The state variable

equation in its modified form can be seen in equation (1.5). It should be noted that the parameter D has been squared due to the substitution $x(t)=w(t)/D$.

$$F(x(t)) = 1-(2x(t)-1)^{2p} \tag{1.4}$$

$$\frac{dx}{dt} = \frac{\eta \mu_D R_{ON}}{D^2} I(t) F(x(t)) \tag{1.5}$$

1.3.2 Biolek Modifications

An alternative window function was proposed in a publication by Zdenek Biolek et al. [17]. When using the windowing function proposed by Joglekar, the motion of the state variable is reduced near the boundary whether it is traveling toward, or away from it. Alternatively, Biolek's window only reduces velocity at the boundary the state variable motion is tending toward. This appears to be a more accurate assumption based on the data collected by HP Labs that was presented in [27]. The Biolek window function is described in equations (1.6) and (1.7).

$$F(x(t)) = 1-(x(t)-stp(-I(t)))^{2p} \tag{1.6}$$

$$stp(I(t)) = \begin{cases} 1 \; if \; I(t) > 0 \\ 0, if \; I(t) < 0 \end{cases} \tag{1.7}$$



Fig. 1.3 Both the Joglekar (left) and the Biolek (right) window functions are plotted where p=1, p=6, and p=100. It can be seen that if a hard limit effect at the borders is desired, then this can be implemented by setting p to a large number in each windowing function.

1.3.3 SPICE Model

The circuit layout for a SPICE model based on equations (1.1) through (1.7) is shown in Fig. 1.4. The two terminals *TE* and *BE* represent the top and bottom electrodes of the memristor. The current source *Gm* generates a current based on equation (1.1). The value of the state variable is determined using a current source and an integrating capacitor. The output of the current source is set equal to the right hand side of equation (1.5), and the value $x(t)$ is determined using the integrating capacitor *Cx*. Memristor SPICE models have been previously proposed using a similar setup [17, 22]. The port *XSV* was created to provide a convenient method for plotting the state variable during a simulation. This is a helpful tool for debugging, but *XSV* should not be used as a voltage output in a circuit design. The circuit schematic in Fig. 1.5 shows how the simulations were performed in LTspice. The memristor was simply connected to an input voltage source. Unless otherwise noted, all simulations in this chapter were performed using this arrangement.

Fig. 1.6 displays code for the memristor subcircuit using the Joglekar window function, and Fig. 1.7 contains the code for the subcircuit using the Biolek window function. The code in Figs. 1.6 and 1.7 is based on the subcircuit proposed in [17], although significant modifications have been made.



Fig. 1.4 Circuit schematic for the memristor SPICE subcircuit based on [2,14,17].



Fig. 1.5 Circuit used to carry out SPICE simulations.

```
* HP Memristor SPICE Model Using Joglekar Window

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* XSV: External connection to plot state variable
*      that is not used otherwise

.SUBCKT MEM_JOGLEKAR TE BE XSV

* Ron: Minimum device resistance
* Roff: Maximum device resistance
* D:   Width of the thin film
* uv:  Dopant mobility
* p:   Parameter for window function
* x0:  State variable initial value

.params Ron=1K Roff=100K x0=.5 D=10N uv=10F p=1

* Joglekar Window Function
.func f(V1) = 1-pow((2*V1-1),(2*p))

* Memristor I-V Relationship
.func IVRel(V1,V2) = V1/(Ron*V2 + Roff*(1-V2))

* Circuit to determine state variable
Gx 0 XSV value={ I(Gmem)*Ron*uv*f(V(XSV,0))/pow(D,2) }
Cx XSV 0 {1}
.ic V(XSV) = x0

* Current source representing memristor
Gmem TE BE value={IVRel(V(TE,BE),V(XSV,0))}

.ENDS MEM JOGLEKAR
```

Fig. 1.6 SPICE subcircuit for the memristor model developed using the Joglekar window function.

```
* HP Memristor SPICE Model Using Biolek Window

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* XSV: External connection to plot state variable
*      that is not used otherwise

.SUBCKT MEM_BIOLEK TE BE XSV

* Ron: Minimum device resistance
* Roff: Maximum device resistance
* D:   Width of the thin film
* uv:  Dopant mobility
* p:   Parameter for window function
* x0:  State varaible initial value

.params Ron=1K Roff=100K x0=.5 D=10N uv=10F p=1

* Biolek Window Function
.func f(V1,I1)={1-pow((V1-stp(-I1)),(2*p))}

* Memristor I-V Relationship
.func IVRel(V1,V2) = V1/(Ron*V2 + Roff*(1-V2))

* Circuit to determine state variable
Gx 0 XSV value={I(Gmem)*Ron*uv*f(V(XSV,0),I(Gmem))/pow(D,2)}
Cx XSV 0 {1}
.ic V(XSV) = x0

* Current source representing memristor
Gmem TE BE value={IVRel(V(TE,BE),V(XSV,0))}

.ENDS MEM_BIOLEK
```

Fig. 1.7 SPICE subcircuit for the memristor model developed using the Joglekar window function.

1.3.4 Simulation Results with Joglekar Window

Figs. 1.8 and 1.9 show the simulation results for the memristor SPICE model when using the Joglekar window function with a sinusoidal input of two different amplitudes. Fig. 1.8 shows the result when the model is driven with the voltage input $v(t)=0.9\sin(2\pi ft)$. The result looks very similar to the model results displayed in Fig. 1.1 because the state variable did not reach the boundaries where the window function has the strongest effect. Fig. 1.8 shows the model response to the voltage input $v(t)=\sin(2\pi ft)$. This slightly larger voltage input forces the value of the state variable into the region affected by the windowing function and harder switching result can be seen.



Fig. 1.8 LTspice simulation results for the memristor model with the Joglekar window function. In this simulation: R_{ON}=100Ω, R_{OFF}=10kΩ, μ_D=5(10^{-14})$m^2s^{-1}V^{-1}$, D=12nm, x_0=0.56, p=7, and $V(t)$=0.9sin(2π10t).



Fig. 1.9 LTspice simulation results for the memristor model with the Joglekar window function and a large input voltage magnitude. In this simulation: R_{ON}=100Ω, R_{OFF}=10kΩ, μ_D=5(10^{-14})$m^2s^{-1}V^{-1}$, D=12nm, x_0=0.56, p=7, and $V(t)$=sin(2π10t).

Fig. 1.10 shows the result using the Joglekar window function when applying several triangular voltage pulses. This result is similar to the one seen in Fig. 1.2, although the most conductive hysteresis loop in this simulation is significantly larger.



Fig. 1.10 Results when simulating the HP Labs memristor with a triangular pulsed input. In this simulation: R_{ON}=1kΩ, R_{OFF}=10kΩ, μ_D=2(10^{-14})$m^2s^{-1}V^{-1}$, D=85nm, x_0=0.093, and p=2. Triangular pulses have magnitude of 1V and 1 second pulse width with rise and fall time of 0.5 seconds.

1.3.5 Simulation Results with Biolek Window

Simulations using the Biolek window function were also performed and the results can be seen in Figs. 1.11 and 1.12. Fig 1.11 shows the model result with a sinusoidal voltage input with large enough amplitude to drive the state variable into the boundary where the window function has a larger impact. Fig. 1.12 displays the result when a triangular pulse input is applied. As opposed to the previous results, this model shows an asymmetric hysteresis with respect to voltage polarity for each of the simulations.



Fig. 1.11 LTspice simulation results for the memristor model with the Joglekar window function. In this simulation: R_{ON}=100Ω, R_{OFF}=1kΩ, μ_D=4(10^{-14})$m^2s^{-1}V^{-1}$, D=16nm, x_0=0.076, p=7, and $V(t)$=sin(2π10t).



Fig. 1.12 Results when simulating the HP Labs memristor with a triangular pulsed input. In this simulation: R_{ON}=1kΩ, R_{OFF}=10kΩ, μ_D=2(10^{-14})$m^2s^{-1}V^{-1}$, D=80nm, x_0=0.1, and p=2. Triangular pulses have magnitude of 1V and 1 second pulse width with rise and fall time of 0.5 seconds.

1.3.6 Discussion

There are several discrepancies when comparing the results obtained from these models to published physical characterization data. When applying a pulsed waveform, the results of these models show that the size of the hysteresis loops in the positive regime increases as conductivity increases. When looking at the published characterization data [7-9], an opposite trend is present.

Also, physical memristor devices show a threshold voltage where hysteresis is not seen unless the voltage across the memristor exceeds the threshold. No threshold voltage is present in these models. Lastly, Matthew D. Pickett et al. at HP Labs published characterization data where the motion of the state variable depends on both its value and the polarity of the applied current [27]. This may suggest that the compression of oxygen vacancies has slightly different dynamics than the oxygen vacancy expansion, and that they are not perfectly mirrored. These models show the motion of the state variable to be equivalent, whether it is moving in the positive or negative direction. The remainder of this chapter discusses the different techniques that have been used to develop memristor SPICE models that provide a closer match to published characterization data.

1.4: Hardware Correlated Models

Alternative memristor models have been proposed that correlate more closely to physical characterization data. This section discusses two different techniques for modeling the memristors [16, 24] where the input voltage is sinusoidal (as opposed to using a repetitive pulse input). Each of these models produces a result that is matched very closely to a specific memristor device.

1.4.1 Air Force Research Lab Model

The first hardware correlated model was developed by Dr. Robinson E. Pino et al. at the Air Force Research Lab (AFRL) to match the I-V characteristic of a device developed at Boise State University [9]. This model matches the IV characteristic very well because it correlates voltage amplitude and slope to a piecewise function matched to the characterization of a physical device. Instead of defining a state variable for this model, the rate of change of the resistance was directly defined using equations (1.8) and (1.9). When the voltage across the memristor is greater than T_h, equation (1.8) is used to determine the change in resistance of the memristor device. When the voltage across the memristor is less than T_l, equation (1.9) is used to determine the change in device resistance. Lastly, when the voltage across the memristor is in the range $T_l \geq V(t) \leq T_h$, there is no change in resistance.

$$\frac{dR}{dt} = \begin{cases} -K_{h1} e^{K_{h2}(V(t)-T_h)}, & R(t) > R_{ON} \\ 0, & R(t) \leq R_{ON} \end{cases} \tag{1.8}$$

$$\frac{dR}{dt} = \begin{cases} K_{l1} e^{K_{l2}(V(t)-T_l)}, & R(t) < R_{OFF} \\ 0, & R(t) \geq R_{OFF} \end{cases} \tag{1.9}$$

The equations for this model were first proposed in [16], and the LTspice code was developed for use in this chapter. The subcircuit can be seen in Fig 1.13. The model has three terminals, again with two representing the top and bottom electrodes of the device and a third terminal to plot the time integral of the rate equation. In this case, the change in resistance is output at the terminal *RSV* since there is no state variable defined other than the resistance itself. It can also be seen that the current through the memristor is determined through a division of two voltages. Even though the value of *V*(*RSV*) is a voltage in the simulation, it actually represents the resistance of the memristor.

```
* Code for memristor model proposed by Dr. Pino et al.

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* RSV: External connection to plot resistance
*      that is not used otherwise

.SUBCKT MEM_PINO TE BE RSV

* Ron:      Minimum device resistance
* Roff:     Maximum device resistance
* Th:       Positive voltage threshold
* Tl:       Negative voltage threshold
* Kh1, Kh2: Fitting params for pos voltage
* Kl1, Kl2: Fitting params for neg voltage

.params Ron=160 Roff=1200 Th=0.2 Tl=-0.35 Kh1=5.5e6 Kh2=-20
+Kl1=4e6 Kl2=20

* Fits the change in resistance to characterization data
.func Rt(V1, V2) = IF( V1 <= Th, IF(V1 >= Tl, 0, IF(V2 <
+Roff, Kl1*exp(Kl2*(V1-Tl)), 0) ), IF(V2 > Ron, -
+Kh1*exp(Kh2*(V1-Th)), 0) )

* Circuit to integrate to find resistance
Gx 0 RSV value={Rt(V(TE,BE),V(RSV))}
Cx RSV 0 {1}
.ic V(RSV) = Roff

* Current source representing memristor
Gmem TE BE value = {V(TE,BE)/V(RSV)}

.ENDS MEM_PINO
```

Fig. 1.13 LTspice subcircuit developed based on the AFRL memristor model equations in [16].

The model result can be seen in Fig. 1.14. When comparing this result to the characterization data in [16], the model matches the both the current output and the I-V curve very closely. For convenience, the physical characterization data that this model was meant to match was reproduced in Fig. 1.15.



Fig. 1.14 Simulation results using the model presented in [16]. In this simulation: R_{ON}=160, R_{OFF}=1200, T_h=0.2, T_l=-0.35, K_{h1}=5.5(10^6), K_{h2}=-20, K_{l1}=4(10^6), and K_{l2}=20.



Fig. 1.15 Plot that displays the characterization data that was used to develop the AFRL model. This figure is a reproduction from [16] that is supplied with permission from the authors.

1.4.2 HP Labs MIM Model

A more complex model was proposed by Drs. Hisham Abdalla and Matthew D. Pickett at HP Labs [24]. This model was based on the assumption that the memristor acts as a metal-insulator-metal (MIM) tunnel barrier [26]. The insulating tunnel barrier is represented by the TiO_2 layer, and the TiO_{2-x} layer acts as a low resistivity metallic layer. As voltage is applied, the thickness of the tunnel barrier is said to modulate due to the position of the oxygen vacancies in the device. This model appears to match the characterization data very well. Additionally, the model is supported by a very strong connection to the physical mechanisms within the device.

The model equations can be seen in (1.10) through (1.14). In these equations the dynamic state variable is defined as $w(t)$. Although, in this model $w(t)$ represents the thickness of the TiO_2 layer as opposed to the TiO_{2-x} layer as seen in equations (1.1) through (1.3). Equations (1.10) and (1.11) govern the state variable

dynamics. These equations were formed based the data presented in [27]. When $I(t) > 0$, the state variable motion is described by equation (1.10), otherwise, the state variable motion is described by equation (1.11). The fitting parameters in the model were defined as follows: f_{off}=3.5 μs, i_{off}=115 μA, a_{off}=1.2 nm, f_{on}=40 μs, i_{on}=8.9 μA, a_{on}=1.8 nm, b=500 μA, and w_c=107 pm.

$$\frac{dw}{dt} = f_{off}\sinh\left(\frac{|I(t)|}{i_{off}}\right)\exp\left(-\exp\left(\frac{w(t)-a_{off}}{wc}-\frac{|I(t)|}{b}\right)-\frac{w(t)}{w_c}\right) \tag{1.10}$$

$$\frac{dw}{dt} = -f_{on}\sinh\left(\frac{|I(t)|}{i_{on}}\right)\exp\left(-\exp\left(\frac{a_{on}-w(t)}{wc}-\frac{|I(t)|}{b}\right)-\frac{w(t)}{w_c}\right) \tag{1.11}$$

Equations (1.12) through (1.14) were developed by modifying the MIM tunnel barrier equations first proposed in [26] to account for a variable barrier width. In these equations: ϕ_0=0.95 V, w_1=0.1261 nm, B=10.24634Δw, λ=0.0998/$w(t)$, and $\Delta w=w_2-w_1$. The variable v_g represents the voltage across the TiO_2 layer of the memristor. The total voltage across the device is equal to the sum of v_g and v_r, where v_r is equal to the voltage across the TiO_{2-x} layer.

$$I(t) = \frac{0.0617}{\Delta w^2}\left\{\phi_I e^{-B\sqrt{\phi_I}} - \left(\phi_I + |v_g|\right)e^{-B\sqrt{\phi_I+|v_g|}}\right\} \tag{1.12}$$

$$\phi_I = \phi_0 - |v_g|\left(\frac{w_1+w_2}{w(t)}\right)-\left(\frac{0.1148}{\Delta w}\right)\ln\left(\frac{w_2(w(t)-w_1)}{w_1(w(t)-w_2)}\right) \tag{1.13}$$

$$w_2 = w_1 + w(t)\left(1-\frac{9.2\lambda}{\left(2.85+4\lambda-2|v_g|\right)}\right) \tag{1.14}$$

The circuit used to test the model differed slightly compared to the original circuit in Fig. 1.7. The circuit in Fig. 1.16 shows how this memristor model was tested. The 2.4 kΩ resistor was added to model the resistance of the electrodes used to characterize the memristor device.

The LTspice code for the MIM memristor SPICE model can be seen in Fig. 1.17. The code was taken from [24], except small changes were made so that the model would operate correctly in LTspice. Additionally, the node *WSV* was added as a terminal so that the state variable motion could be plotted easily.



Fig. 1.16 Schematic for testing the MIM memristor model [24].

```
* HP Labs MIM Model proposed by Drs. Abdalla and Pickett

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* WSV: External connection to plot state variable
*      that is not used otherwise

.SUBCKT MEM_ABDALLA TE BE WSV

.params phio=0.95 Lm=0.0998 w1=0.1261 foff=3.5e-6 ioff=115e-6
+aoff=1.2 fon=40e-6 ion=8.9e-6 aon=1.8 b=500e-6 wc=107e-3
+winit = 1.2

* MIM IV Relationship
G1 TE internal val-
ue={sgn(V(x))*pow((1/V(dw)),2)*0.0617*(V(phiI)*exp(-
+V(B)*V(sr))-(V(phiI)+abs(V(x)))*exp(-V(B)*V(sr2)))}
Esr sr 0 value={sqrt(V(phiI))}
Esr2 sr2 0 value={sqrt(V(phiI)+abs(V(x)))}
* Series resistance of TiO2-x Layer
Rs internal BE 215
Eg x 0 value={V(TE)-V(internal)}
Elamda Lmda 0 value={Lm/V(W)}
Ew2 w2 0 value={w1+V(WSV)-(0.9183/(2.85+4*V(Lmda)-
+2*abs(V(x))))}
EDw dw 0 value={V(w2)-w1}
EB B 0 value={10.246*V(dw)}
ER R 0 value={(V(w2)/w1)*(V(WSV)-w1)/(V(WSV)-V(w2))}
EphiI phiI 0 value={phio-abs(V(x))*((w1+V(w2))/(2*V(WSV)))-
+1.15*V(Lmda)*V(WSV)*log(V(R))/V(dw)}
* State variable integrating capactior
C1 WSV 0 1e-9
.ic V(WSV)=winit
R WSV 0 1e8MEG
Ec c 0 value={abs(V(internal)-V(BE))/215}
Emon1 mon1 0 value={((V(WSV)-aoff)/wc)-(V(c)/b)}
Emon2 mon2 0 value={(aon-V(WSV))/wc-(V(c)/b)}
Goff 0 WSV value={foff*sinh(stp(V(x))*V(c)/ioff)*exp(-
+exp(V(mon1))-V(WSV)/wc)}
Gon WSV 0 value={fon*sinh(stp(-V(x))*V(c)/ion)*exp(-
+exp(V(mon2))-V(WSV)/wc)}

.ENDS MEM_ABDALLA
```

Fig. 1.17 LTspice code for the HP Labs MIM memristor model [24].

The simulation results for the SPICE model can be seen in Fig. 1.18. The top left plot shows the voltage across the memristor along with the current through the memristor. The voltage signal from the input source can be seen in the bottom left plot. The simulated I-V curve closely matches the data displayed in [24]. For convenience, the characterization data from [24] is displayed alongside the model result in Fig. 1.19.



Fig. 1.18 Simulation results using the HP Labs MIM model.



Fig. 1.19 Memristor characterization data from [24] matched to the HP Labs MIM model. This figure was reproduced from [24] with permission from the authors.

1.4.3 Discussion

The models presented in this section show a very close correlation to the characterizations that they were designed to match. The disadvantage is that it is not known how closely these models will match the characterization data for alternative voltage inputs such as repetitive triangular pulses. Also, these models are very specific to a single fabricated device. Given the wide variety in the current voltage characterization of memristors, these models will most likely have to be updated significantly for use with different device structures.

1.5: Hyperbolic Sine Models

The hyperbolic sinusoid shape has been proposed for several memristor models [13, 15, 18, 23]. This is because the hyperbolic sinusoid function can be used to approximate the I-V relationship of an MIM junction [26]. Since thinfilm memristors are commonly fabricated by sandwiching an oxide between two metal electrodes, modeling a memristor as an MIM device seems reasonable. Section 1.4.2 has already demonstrated a model [24] based on MIM tunneling equations. Using a hyperbolic sine function in the I-V relationship appears provide a significantly better result when using a repetitive voltage pulse input.

1.5.1 General Hyperbolic Sine Model

A general hyperbolic sinusoid model was proposed by Dr. Mika Laiho et al. [15] and is described in equations (1.15) and (1.16). The current voltage relationship is represented by a hyperbolic sinusoid modulated by the value of the state variable $x(t)$. The parameters a_1, a_2, b_1, and b_2 are used to adjust the I-V response of the model. The state variable is also modeled using a hyperbolic sine function (see equation (1.16)). Several memristor characterizations show that the state of the device will not change unless the voltage applied exceeds a threshold [7-12], and the hyperbolic sinusoid based state variable is one option that achieves this effect. The constants c_1, c_2, d_1, and d_2, are used to shape the threshold and intensity of the state variable dynamics.

$$I(t) = \begin{cases} a_1 x(t) \sinh\big(b_1 V(t)\big), & V(t) \geq 0 \\ a_2 x(t) \sinh\big(b_2 V(t)\big), & V(t) < 0 \end{cases} \tag{1.15}$$

$$\frac{dx}{dt} = \begin{cases} c_1 \sinh\big(d_1 V(t)\big), & V(t) \geq 0 \\ c_2 \sinh\big(d_2 V(t)\big), & V(t) < 0 \end{cases} \tag{1.16}$$

This model was originally proposed in [15] without boundary conditions to stop the state variable from exceeding the range $0 \leq x(t) \leq 1$. To address this issue, the model has been modified by using the Biolek window function [17] to define the device boundaries. The updated state variable equation can be seen in (1.17) where $F(x(t))$ represents the Biolek window function. The SPICE code for the model with and without the boundary addition has been developed and can be seen in Figs. 1.20 and 1.21 respectively.

$$\frac{dx}{dt} = \begin{cases} c_1 \sinh\big(d_1 V(t)\big) F\big(x(t)\big), & V(t) \geq 0 \\ c_2 \sinh\big(d_2 V(t)\big) F\big(x(t)\big), & V(t) < 0 \end{cases} \tag{1.17}$$

```
* SPICE model for equations proposed by Dr. Mika Laiho et al.

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* XSV: External connection to plot state variable
*      that is not used otherwise

.subckt MEM_LAIHO TE BE XSV

.param a1=4e-8 b1=1.2 a2=1.25e-7 b2=1.2 c1=6e-4 d1=2
+c2=6.6e-4 d2=3.8 x0=0.001

* Hyperbolic sine IV relationship
.func IVRel(V1,V2) = IF(V1 >= 0, a1*V2*sinh(b1*V1),
+a2*V2*sinh(b2*V1))

* Equation for state variable
.func SV(V1) = IF(V1 >= 0, c1*sinh(d1*V1), c2*sinh(d2*V1))

* Current source representing memristor
Gmem TE BE value = {IVRel(V(TE,BE),V(XSV,0))}

* Circuit to determine value of state variable
Gxsv 0 XSV value = {SV(V(TE,BE))}
Cx XSV 0 {1}
.ic V(XSV) = x0

.ends MEM_LAIHO
```

Fig. 1.20 LTspice code that was developed for the generalized hyperbolic sinusoid model proposed by Laiho et al.

Figs. 1.22 and 1.23 show the simulation results for this model with and without the addition of the Biolek window function. Fig. 1.22 shows results of the model as it was presented in [15]. Modeling the boundary is not an issue since not enough charge was applied to let the state variable move outside the boundaries. Fig. 1.23 shows the results of the model where the Biolek window function was added to set boundaries on the value of the state variable. It can be seen that current through the device increases with each voltage pulse until the upper limit of the state variable motion is achieved. At this point the peak of each current pulse no longer changes until the polarity of the input voltage is reversed.

```
* SPICE model for equations proposed by Dr. Mika Laiho et al.
* with Biolek window for memristor boundaries

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* XSV: External connection to plot state variable
*      that is not used otherwise

.subckt MEM_LAIHO_WINDOW TE BE XSV

.param a1=4e-8 b1=1.2 a2=1.25e-7 b2=1.2 c1=6e-4 d1=2
+c2=6.6e-4 d2=3.8 x0=0.001 p=1

* Hyperbolic sine IV relationship
.func IVRel(V1,V2) = IF(V1 >= 0, a1*V2*sinh(b1*V1),
+a2*V2*sinh(b2*V1))

* Equation for state variable
.func SV(V1) = IF(V1 >= 0, c1*sinh(d1*V1), c2*sinh(d2*V1))

* Biolek window function
.func f(V1,I1)={1-pow((V1-stp(-I1)),(2*p))}

* Current source representing memristor
Gmem TE BE value = {IVRel(V(TE,BE),V(XSV,0))}

* Circuit to determine value of state variable
Gxsv 0 XSV value = {SV(V(TE,BE))*f(V(XSV,0),I(Gmem))}
Cx XSV 0 {1}
.ic V(XSV) = x0

.ends MEM_LAIHO_WINDOW
```

Fig. 1.21 LTspice code that was developed for the generalized hyperbolic sinusoid model with the addition of the Biolek windowing function.



Fig. 1.22 Simulation results for the hyperbolic sinusoid model proposed by Laiho et al. In this simulation: a_1=4(10^{-8}), b_1=1.2, a_2=1.25(10^{-7}), b_2=1.2, c_1=6(10^{-4}), d_1=2, c_2=6.6(10^{-4}), d_2=3.8, and x_0=0.001. Triangular pulses have magnitude of +5V/-2.5V and a 0.1 second pulse width with rise and fall time of 0.05 seconds.



Fig. 1.23 Simulation results for the hyperbolic sinusoid model proposed by Laiho et al. with the addition of the Biolek window function. In this simulation: a_1=4(10^{-8}), b_1=1.2, a_2=1.25(10^{-7}), b_2=1.2, c_1=6(10^{-4}), d_1=2, c_2=6.6(10^{-4}), d_2=3.8, and p=1, x_0=0.001. Triangular pulses have magnitude of +5.5V/-3V and a 0.1 second pulse width with rise and fall time of 0.05 seconds.

1.5.2 University of Michigan Model

An alternative model based on the hyperbolic sine I-V relationship was developed by Ting Chang et al. in [23], and the corresponding SPICE code can be seen in Fig. 1.24. The I-V relationship can be seen in (1.18) where the first term is due to a Schottky barrier between the oxide layer and the bottom electrode, and the second term is due to the tunneling through the MIM junction. The state variable $x(t)$ is a value between 0 and 1 that represents the ion migrations which determine the conductivity of the device. The motion of the state variable is described by equation (1.19) which is similar to equation (1.16). The fitting parameters η_1, η_2, and λ are used shape the dynamics of the state variable equation. In each of the simulations in Figs. 1.25 through 1.27, the constants in the equations were defined as follows: α=5(10^{-7}), β=0.5, γ=4(10^{-6}), δ=2, λ=4.5, η_1=0.004, η_2=4, and τ=10.

$$I(t) = (1 - x(t))\alpha\left[1 - e^{\beta V(t)}\right] + x(t)\gamma \sinh(\delta V(t)) \tag{1.18}$$

$$\frac{dx}{dt} = \lambda\left[\eta_1 \sinh(\eta_2 V(t))\right] \tag{1.19}$$

An alternative function for the motion of the state variable was proposed in [23] to account for the overlapping of multiple hysteresis loops where the device was tested with a repetitive pulse input (see Fig 1.28). The overlap in hysteresis was said to be caused by diffusion of ions within the device. Equation (1.20) shows the modified state variable equation with the diffusion term added. The LTspice code for the model was obtained from [23], and was then modified so that either state variable equation could be used with the change of a binary variable. This allows for the drift component to be turned on or off from the simulation that

is using the subcircuit. The resulting SPICE subcircuit can be seen in Fig. 1.24. It should be noted that equations (1.19) and (1.20) were taken from the SPICE code in [23], not from the text in [23]. The equations in the text differed slightly so precedence was given to the equations used to develop the model.

$$\frac{dx}{dt} = \lambda\left[\eta_1 \sinh(\eta_2 V(t)) - \frac{x(t)}{\tau}\right] \tag{1.20}$$

```
* Memristor subcircuit developed by Chang et al.

* Connections:
* TE:  Top electrode
* BE:  Bottom electrode
* XSV: External connection to plot state variable
*      that is not used otherwise

.SUBCKT MEM_CHANG TE BE XSV

* Parameters:
* alpha:      Prefactor for Schottky barrier
* beta:       Exponent for Schottky barrier
* gamma:      Prefactor for tunneling
* delta:      Exponent for tunneling
* xmax:       Maximum value of state variable
* xmin:       Minimum value of state variable
* drift_bit:  Binary value to switch the ionic drift in (1)
*             or out (0) of the equation
* lambda:     State variable multiplier
* eta1, eta2: State variable exponential rates
* tau:        Diffusion coefficient

.param alpha=0.5e-6 beta=0.5 gamma=4e-6 delta=2 xmax=1 xmin=0
+drift_bit = 0 lambda=4.5 eta1=0.004 eta2=4 tau=10

.param cp={1}
Cpvar XSV 0 {cp}

* Rate equation for state variable
Gx 0 XSV value={
trunc(V(TE,BE),cp*V(XSV))*lambda*(eta1*sinh(eta2*V(TE,BE))-
+drift_bit*cp*V(XSV)/tau) }

.ic V(XSV) = 0.0

* Auxiliary functions to limit the range of x
.func sign2(var) {(sgn(var)+1)/2}
.func trunc(var1,var2) {sign2(var1)*sign2(xmax-var2)+sign2(-
+var1)*sign2(var2-xmin)}

* Memristor IV Relationship
Gm TE BE value={(1-cp*V(XSV))*alpha*(1-exp(-
+beta*V(TE,BE)))+(cp*V(XSV))*gamma*sinh(delta*V(TE,BE))}

.ENDS MEM_CHANG
```

Fig. 1.24 LTspice code for the for the memristor model proposed by Chang et al. in [23].

The simulation in Fig. 1.25 shows the model results when a voltage signal with zero net charge is applied to the memristor device. The result is similar to previous simulations where a sinusoidal input is applied. In this simulation a prominent curvature can be seen due to the hyperbolic sine term in the I-V relationship. Fig. 1.26 shows the model results when repetitive pulses are applied to the device. The multiple-hysteresis pattern looks similar to the result in Fig. 1.22 although this model shows a higher conductivity when negatively biased. Fig. 1.27 shows the model result when the ion diffusion term was added to the state variable equation. This result appears to be a better match to the characterization data in [23], which was reproduced for convenience in Fig. 1.28. The hysteresis loops overlap in the positively biased area, but a larger gap can be seen between the loops when negative pulses are applied. Ion diffusion appears to be a logical explanation for this effect.



Fig. 1.25 Simulation results for the memristor SPICE model proposed by Chang et al. without the ion diffusion term included in the state variable equation. The triangular pulses have a magnitude of 1.25V and a pulse width of 0.5 seconds with a 0.25 second rise/fall time.



Fig. 1.26 Simulation results for the memristor SPICE model proposed by Chang et al. without the ion diffusion term included in the state variable equation. The triangular pulses have a magnitude of 1.25V and a pulse width of 0.5 seconds with a 0.25 second rise/fall time.



Fig. 1.27 Simulation results for the memristor SPICE model proposed by Chang et al. with the ion diffusion term included in the state variable equation. The triangular pulses have a magnitude of 1.25V and a pulse width of 0.5 seconds with a 0.25 second rise/fall time.



Fig. 1.28 Plot that displays the characterization data that the model in Fig. 1.24 was meant to match. This figure is a reproduction from [23] that was supplied with permission from the authors.

1.5.3 Discussion

The models described in this section use a hyperbolic sine function in the I-V relationship which models the characteristics of a memristor well for both single sweep and repetitive pulse inputs. Additional properties were also modeled such as the Schottky barrier at a metal-oxide interface, and the diffusion of ions. This resulted in a stronger correlation to physical memristor characterization data. These models have the potential to describe the functionality of a memristor in a more generalized way which also appears to be quite accurate. The drawback is that these models do not correlate to physical hardware as closely as the models in Section 1.4. The next section describes a generalized model that quantitatively matches published characterization data for a variety of different memristor devices for a variety of different voltage inputs.

1.6: Generalized Model for Many Devices

A memristor device model was developed in [13] that can accurately match the I-V characteristic of several published memristor devices. The equations were developed based on a more general understanding of memristor dynamics, and fitting parameters were used to match the results to physical characterization data [7-12].

1.6.1 Generalized Memristor Model Equations

The generalized I-V relationship for this memristor model can be seen in equation (1). A similar equation was proposed in [15] that used two separate multiplying parameters in the hyperbolic sine term depending on voltage polarity. In the development of this model, it was determined that a single parameter b, could be used independent of voltage polarity. The hyperbolic sinusoid shape is due to the MIM structure [26] of memristors, which causes the device to have an increase in conductivity beyond a certain voltage threshold. The parameters a_1, a_2, and b are used to fit equation (1.21) to the different device structures of the memristors studied in this paper. Based on existing memristor characterization data, the devices appear to be more conductive in the positive region. To account for this, a different amplitude parameter is required depending on the polarity of the input voltage. The fitting parameter b was used to control the intensity of the threshold function relating conductivity to input voltage magnitude. For example, the device published in [7] has a stronger threshold (b=3) than the device published in [8] (b=0.7).

The I-V relationship also depends on the state variable $x(t)$, which provides the change in resistance based on the physical dynamics in each device. In this model, the state variable is a value between 0 and 1 that directly impacts the conductivity of the device.

$$I(t) = \begin{cases} a_1\, x(t)\sinh\big(bV(t)\big), & V(t) \geq 0 \\ a_2\, x(t)\sinh\big(bV(t)\big), & V(t) < 0 \end{cases} \tag{1.21}$$

The change in the state variable is based on two different functions, namely, $g(V(t))$ and $f(x(t))$. The function $g(V(t))$ imposes a programming threshold on the memristor model. The threshold is viewed as the minimum energy required to alter the physical structure of the device. Each of the published memristor devices [7-12] show that there is no state change in the memristor unless a certain voltage threshold is exceeded. These changes include the motion of low mobility ions or dopants [7, 8, 10-12], or the state change in a chalcogenide device [9]. The programming threshold was implemented using equation (1.22). As opposed to the hyperbolic sinusoid programming threshold implemented in [15], the method in

equation (1.22) provides the possibility of having different thresholds based on the polarity of the input voltage. This is required to provide a better fit to the characterization data, since several of these devices show different threshold values depending on whether the input voltage is positive or negative.

In addition to the positive and negative thresholds (V_p and V_n), the magnitude of the exponentials (A_p and A_n) can be adjusted. The magnitude of the exponential represents how quickly the state changes once the threshold is surpassed. In the results in Section IV, it can be seen that the chalcogenide device [9] requires a very large change once the threshold is surpassed. Alternatively, the device based on the motion of silver dopants [8] requires a much lower amplitude coefficient as this appears to be a slower phenomenon.

$$g(V(t)) = \begin{cases} A_p\left(e^{V(t)} - e^{V_p}\right), & V(t) > V_p \\ -A_n\left(e^{-V(t)} - e^{V_n}\right), & V(t) < -V_n \\ 0, & -V_n \le V(t) \le V_p \end{cases} \tag{1.22}$$

The second function used to model the state variable *f(x(t))*, can be seen in (1.23) and (1.24). This function was added based on the assumption that it becomes harder to change the state of the devices as the state variable approaches the boundaries. This idea was theorized in [14, 17], and demonstrated experimentally in [27]. Also, this function provides the possibility of modeling the motion of the state variable differently depending on the polarity of the input voltage. This is a necessary addition as it has been experimentally verified that the state variable motion is not equivalent in both directions [27]. The memristor device model published in [24] also uses switching state variable where the motion varies depending on the polarity of the current through the device. One possible explanation for this may be that it is more difficult to put ions back in their original position after they have been previously moved. When $\eta V(t)>0$, the state variable motion is described by equation (1.23), otherwise the motion is described by (1.24). The term η was introduced to represent the direction of the motion of the state variable relative to the voltage polarity. When η=1, a positive voltage (above the threshold) will increase the value of the state variable, and when η=–1, a positive voltage results in a decrease in state variable. A similar technique was introduced in [14].

The function *f(x(t))* was developed by assuming the state variable motion was constant up until the point x_p or x_n. At this point the motion of the state variable was limited by a decaying exponential function. Since the motion of the state variable appears to be different across the different types of devices studied, this function used fitting parameters to accommodate the variety. The constants in this equation represent the point where the state variable motion becomes limited (x_p and x_n), and the rate at which the exponential decays (α_n and α_p). These differences may be due to the fact that the motion of the state change in a chalcogenide device is very different than the motion of ions or dopants.

$$f(x(t)) = \begin{cases} e^{-\alpha_p(x(t)-x_p)} w_p(x(t), x_p), & x(t) \geq x_p \\ 1, & x(t) < x_p \end{cases} \quad (1.23)$$

$$f(x(t)) = \begin{cases} e^{\alpha_n(x(t)+x_n-1)} w_n(x(t), x_n), & x(t) \leq 1 - x_n \\ 1, & x(t) > 1 - x_n \end{cases} \quad (1.24)$$

In equation (1.25), $w_p(x,x_p)$ is a windowing function that ensures $f(x)$ equals zero when $x(t)$=1. In equation (1.26), $w_n(x,x_n)$ keeps $x(t)$ from becoming less than 0 when the current flow is reversed.

$$w_p(x, x_p) = \frac{x_p - x}{1 - x_p} + 1 \quad (1.25)$$

$$w_n(x, x_n) = \frac{x}{1 - x_n} \quad (1.26)$$

Equation (1.27) is used to model the state variable motion in each of the memristor devices. Since the modeled state variable must match devices with many different physical structures, this equation is very different than the equation in [2] that was used to model only TiO_2 devices. The term η is also used in (1.27) to determine the direction of the dynamic state variable motion.

$$\frac{dx}{dt} = \eta g(V(t)) f(x(t)) \quad (1.27)$$

1.6.2 SPICE Code for Generalized Memristor Model

The SPICE code for this memristor model can be seen Fig. 1.29. The circuit structure is equivalent to the one displayed in Fig 1.4, although the equations for $I(t)$ and $I_{Gx}(t)$ differ in this model.

```
* Memristor model for many devices

* Connections:
* TE - top electrode
* BE - bottom electrode
* XSV - External connection to plot state variable
* that is not used otherwise

.subckt MEM_YAKOPCIC TE BE XSV

* Fitting parameters to model different devices
* a1, a2, b:      Parameters for IV relationship
* Vp, Vn:         Pos. and neg. voltage thresholds
* Ap, An:         Multiplier for SV motion intensity
* xp, xn:         Points where SV motion is reduced
* alphap, alphan: Rate at which SV motion decays
* xo:             Initial value of SV
* eta:            SV direction relative to voltage

.params a1=0.17 a2=0.17 b=0.05 Vp=0.16 Vn=0.15 Ap=4000
+An=4000 xp=0.3 xn=0.5 alphap=1 alphan=5 xo=0.11 eta=1

* Multiplicitive functions to ensure zero state
* variable motion at memristor boundaries
.func wp(V) = (xp-V)/(1-xp)+1
.func wn(V) = V/(1-xn)

* Function G(V(t)) - Describes the device threshold
.func G(V) = IF(V <= Vp, IF(V >= -Vn, 0, -An*(exp(-
+V)-exp(Vn))), Ap*(exp(V)-exp(Vp)))

* Function F(V(t),x(t)) - Describes the SV motion
.func F(V1,V2) = IF(eta*V1 >= 0, IF(V2 >= xp, exp(-
+alphap*(V2-xp))*wp(V2) ,1), IF(V2 <= (1-xn),
+exp(alphan*(V2+xn-1))*wn(V2) ,1))

* IV Response - Hyperbolic sine due to MIM structure
.func IVRel(V1,V2) = IF(V1 >= 0, a1*V2*sinh(b*V1),
+a2*V2*sinh(b*V1) )

* Circuit to determine state variable
* dx/dt = F(V(t),x(t))*G(V(t))
Cx XSV 0 {1}
.ic V(XSV) = xo
Gx 0 XSV value={eta*F(V(TE,BE),V(XSV,0))*G(V(TE,BE))}
* Current source for memristor IV response
Gm TE BE value = {IVRel(V(TE,BE),V(XSV,0))}

.ends MEM_YAKOPCIC
```

Fig. 1.29 LTspice subcircuit for the generalized memristor device model proposed in [13].

1.6.3 Generalized Memristor Model Results

To display the functionality of this SPICE model, each of the published memristor devices [7-12] was modeled in a simple I-V simulation. The circuit used to test the model was equivalent to the one displayed in Fig. 1.7.

Figs. 1.30 and 1.31 show the simulation results of the model when it was used to match the characterization data published in [9]. The characterization data from [9] was reproduced in Fig. 1.32 for convenience. Fig. 1.30 displays the first simulation result where the device in [9] was modeled with a sinusoidal input both at 100 Hz and 100 kHz. The hysteresis in the model diminished when the frequency was increased to 100 kHz just as it did in [9]. The simulated I-V characteristic was matched to the 100 Hz data provided in [9] using selected data points (dots in the I-V curve in Fig. 1.30) with an average error of 84.8μA (6.64%). The percent error was determined by first calculating the sum of the differences between the model output current, and the current at each of the selected data point from the characterization. This value was then divided by the sum of the currents at each of the selected points to determine relative error.

Fig. 1.31 shows the simulation result when matching the model to the repetitive pulse input data also provided in [9]. In this case, the model was able match the characterization data with an average error of 32.7μA (6.66%). The characterization data provided for this simulation was only available for when positive voltage sweeps were applied. It was assumed that the parameters for the negative regime would closely match the parameters used in the sinusoidal simulation in Fig. 1.30. The one exception was that the negative conductivity parameter a_2 was set to the values decided for a_1 in the pulsed simulation.



Fig. 1.30 Simulation results when modeling the device in [9] for a sinusoidal input (dots in I-V curve show target data points). The plots show the current and voltage waveforms and the I-V curve. The I-V curve for a high frequency input where the device behaves as a linear resistor is also displayed. In this simulation: V_p=0.16V, V_n=0.15V, A_p=4000, A_n=4000, x_p=0.3, x_n=0.5, α_p=1, α_n=5, a_1=0.17, a_2=0.17, b=0.05, x_0=0.11, and η=1.



Fig. 1.31 Results obtained for matching the pulsed input characterization in [9]. The plots again show the voltage and current waveforms as well as the I-V curve. Dots in the I-V curve show the points from [9]. In this simulation: V_p=0.16V, V_n=0.15V, A_p=4000, A_n=4000, x_p=0.3, x_n=0.5, α_p=1, α_n=5, a_1=0.097, a_2=0.097, b=0.05, x_0=0.001, and η=1.



Fig. 1.32 Characterization data to which the model was matched in Figs. 1.30 and 1.31. Each of these plots showing physical characterization data were reproduced from [14] with permission from the authors.

When comparing the many parameters used to model the device characterized in [9], it can be seen that very few adjustments were necessary when switching between the two modes of operation. Out of the total 12 parameters, 9 of them remained the same between the simulations in Figs. 1.30 and 1.31. The three parameters that were changed include the conductivity parameters a_1 and a_2, and the initial position of the state variable x_0. Each of these parameters is highly related to the device thickness, and these differences could have been caused by non-uniformities in the wafer.

The simulation in Fig. 1.33 was based on the characterization data provided in [10] where cyclic voltage sweeps were applied to the memristor device. The average error in this case was determined to be 8.63μA (13.6%). The error dropped to 8.72% when not considering the largest outlier. The largest discrepancy in this simulation was caused by the lack of curvature in the model when the device was in a conductive state. For convenience, the characterization data that was published in [10] has been reproduced in Fig. 1.34.



Fig. 1.33 Simulation results when modeling the device in [10] for a cyclical DC sweep where the plots show the voltage and current waveforms and the I-V curve (dots in the I-V curve show target data points). V_p=1.2V, V_n=0.6V, A_p=5, A_n=30, x_p=0.7, x_n=0.8, α_p=4, α_n=24, a_1=2.3(10^{-4}), a_2=3.8(10^{-4}), b=1, x_0=0.02, and η=1.



Fig. 1.34 Characterization data to which the model was matched to produce the result in Fig. 1.33. The plot was reproduced from [10] with permission from the authors.

The simulation results for the device characterized in [7] can be seen in Fig. 1.35. In this case the characterization was done using repetitive pulses as opposed to a cyclic voltage sweep. The model was able to match the selected data points from the characterization data with an average error of 1.89μA (11.66%). In Fig. 1.35, it can be seen that several of the repetitive pulses have different peak amplitudes throughout the simulation. This was done to better match the characterization data, as the I-V curve published in [7] also had varying peak voltages (see Fig. 1.36).



Fig. 1.35 Results obtained for matching the device characterization in [7]. The plots show the voltage and current waveforms and the I-V curve where the dots in the I-V curve show target points from [7]. In this simulation: V_p=0.9V, V_n=0.2V, A_p=0.1, A_n=10, x_p=0.15, x_n=0.25, α_p=1, α_n=4, a_1=0.076, a_2=0.06, b=3, x_0=0.001, η=1.



Fig. 1.36 Characterization data from HP Labs to which the memristor model was matched to produce the result in Fig. 1.35. This figure was reproduced from [7] with permission from the author.

The memristor simulation in Fig. 1.37 was based on the device characterized in [8]. This device was characterized in [8] using a slower pulse train that required about 20 seconds to complete. The simulation matches each target data point with an average error of 20.0nA (6.21%). Fig. 1.38 displays a reproduction of the characterization data from [8] to which the model was matched.



Fig. 1.37 Results obtained for matching characterization in [8]. Dots show the points from [8]. In this simulation: V_p=2.1V, V_n=0.8V, A_p=0.03, A_n=0.08, x_p=0.3, x_n=0.5, α_p=1, α_n=3, a_1=1.59(10^{-7}), a_2=2.15(10^{-7}), b=0.7, x_0=0.2, η=1.



Fig. 1.38 Physical device characterization data published in [8]. This figure was reproduced with permission from the authors.

The plots in Fig. 1.39 correspond to the memristor developed in [11, 12]. The simulated I-V characteristic was matched to target data points with an average error of 5.97%. The input voltage waveform was replicated based on the data provided in [12]. The IV characteristic in [11, 12] shows 3 sequential voltage sweeps. A reproduction of this I-V characteristic obtained from [11] can be seen in Fig 1.40. Fig. 1.39 shows the results when modeling target data from the third sweep since the decay in the first two sweeps is most likely due to initial forming and would not be present over a large number of cycles. Contrary to the previous simulations, this device was characterized so that the device conductivity decreases as positive voltage is applied. To accommodate for this, the variable η was set to –1.



Fig. 1.39 Results obtained for matching the characterization in [11, 12]. The plots again show the voltage and current waveforms as well as the I-V curve. Dots in the I-V curve show the points from [11, 12]. In this simulation: V_p=0.65V, V_n=0.56V, A_p=16, A_n=11, x_p=0.3, x_n=0.5, α_p=1.1, α_n=6.2, a_1=1.4, a_2=1.4, b=0.05, x_0=0.99, and η= –1.



Fig. 1.40 I-V characteristic published in [12]. This figure was reproduced with permission from the authors.

1.7 Conclusion

When comparing all of the models, it can be seen that there have been several different techniques proposed for modeling memristor devices. Each of the techniques was validated based on either the matching of the published characterization data, or by modeling behaviors observed in memristor devices.

The models proposed in Section 1.3 were based on directly relating the ionic drift in the oxide layer to the overall device resistance. These models provide a simple explanation of memristor behavior that relate very closely to the theory first proposed by Dr. Chua. Although, these models appear to have the least in common with the published characterization data for different memristor devices.

The models in Section 1.4 show a very close correlation to the characterization data of a specific memristor device, although little is known about how well these models function for alternative device structures and voltage inputs.

The models in Section 1.5 show how using the hyperbolic sine function in the IV relationship provides a simple and effective means for modeling the MIM junction found within a memristor device. These models appear to match memristor behavior especially well when repetitive pulsed inputs are applied, but these models have not been numerically correlated to any fabrication data.

Section 1.6 provides a more generalized SPICE model that is also capable matching the I-V characteristic of several different devices. The disadvantage of this model is that it has less theoretical correlation to the physical mechanisms governing the device when compared to the HP Labs MIM model or the University of Michigan model.

The goal of this chapter was to review existing memristor modeling techniques for use in SPICE simulations and circuit design. The models that were discussed in this chapter can be implemented in LTspice using the subcircuits provided. This allows for a more standardized comparison of memristor models that can all be implemented in a single SPICE program.

Based on the results, it is the authors' recommendation that the model [13] discussed in section 1.6 be used for the most accurate representation of published memristor current-voltage data. It has been shown that this model can be applied to several different materials and device structures. This will become useful in the future as fabrication of memristor devices is continually changing. The fitting parameters of this model could most likely be changed to accommodate for future fabrication techniques, as it has been shown that this model can accurately match the characterizations of a variety of memristors.

References

[1] L. O. Chua, Leon O, "Memristor—The Missing Circuit Element," *IEEE Transactions on Circuit Theory*, 18(5), 507–519 (1971).

[2] D. B. Strukov, G. S. Snider, D. R. Stewart, and R. S. Williams, "The missing Memristor found," *Nature*, 453, 80–83 (2008).

[3] R. Williams, "How We Found The Missing Memristor," *IEEE Spectrum*, vol. 45, no. 12, pp. 28-35, 2008.

[4] T. Raja and S. Mourad, "Digital Logic Implementation in Memristor-based Crossbars," International Conference on Communications, Circuits, and Systems, pp. 939-943, (2009).

[5] E. Lehtonen and M. Laiho, "Stateful Implication Logic with Memristors," IEEE/ACM International Symposium on Nanoscale Architectures, pp. 33-36, (2009).

[6] S. Wald, J. Baker, M. Mitkova, and N. Rafla, "A non-volatile memory array based on nano-ionic Conductive Bridge Memristors," IEEE Workshop on Microelectronics and Electron Devices, pp. 1-4, (2011).

[7] G. S. Snider, "Cortical Computing with Memristive Nanodevices," *SciDAC Review*, (2008).

[8] S. H. Jo, T. Chang, I. Ebong, B. B. Bhadviya, P. Mazumder, and W. Lu, "Nanoscale Memristor Device as Synapse in Neuromorphic Systems," *Nano Letters*, 10 (2010).

[9] A. S. Oblea, A. Timilsina, D. Moore, and K. A. Campbell, "Silver Chalcogenide Based Memristor Devices," *IJCNN*, (2010).

[10] J. J. Yang, M. D. Pickett, X. Li, D. A. A. Ohlberg, D. R. Stewart and R. S. Williams, "Memristive switching mechanism for metal/oxide/metal nanodevices," *Nature Nanotechnology*, 3, 429–433 (2008).

[11] K. Miller, K. S. Nalwa, A. Bergerud, N. M. Neihart, and S. Chaudhary, “Memristive Behavior in Thin Anodic Titania,” *IEEE Electron Device Letters* 31(7), (2010).
[12] K. Miller, “Fabrication and modeling of thin-film anodic titania memristors,” *Master's Thesis*, Iowa State University, Electrical and Computer Engineering (VLSI), Ames, Iowa, (2010).
[13] C. Yakopcic, T. M. Taha, G. Subramanyam, R. E. Pino, S. Rogers, “A Memristor Device Model,” *IEEE Electron Device Letters,* (Accepted for publication).
[14] Y. N. Joglekar and S. J. Wolf, "The elusive memristor: properties of basic electrical circuits," *European Journal of Physics*, 30(661), (2009).
[15] M. Laiho, E. Lehtonen, A. Russel, and P. Dudek, “Memristive synapses are becoming a reality,” *The Neuromorhic Engineer*, (2010).
[16] R. E. Pino, J. W. Bohl, N. McDonald, B. Wysocki, P. Rozwood, K. A. Campbell, A. Oblea, and A. Timilsina, “Compact method for modeling and simulation of memristor devices: Ion conductor chalcogenide-based memristor devices,” IEEE/ACM International Symposium on Nanoscale Architectures, pp. 1 – 4, (2010).
[17] Z. Biolek, D. Biolek, V. Biolková, "Spice Model of Memristor with Nonlinear Dopant Drift," *Radioengineering*, 18(2), 210-214 (2009).
[18] E. Lehtonen and M. Laiho, “CNN using memristors for neighborhood connections,” feb. 2010, pp. 1 –4.
[19] D. Batas, H. Fiedler, “A Memristor SPICE Implementation and a New Approach for Magnetic Flux-Controlled Memristor Modeling,” *IEEE Transactions on Nanotechnology*, 10(2), 2011.
[20] A. Rak and G. Cserey, “Macromodeling of the memristor in spice,” *Computer-Aided Design of Integrated Circuits and Systems, IEEE Transactions on*, vol. 29, no. 4, pp. 632 –636, Apr. 2010.
[21] S. Benderli and T. Wey, “On SPICE macromodelling of TiO2 memristors,” *Electronics Letters*, vol. 45, no. 7, pp. 377–379, 2009.
[22] M. Mahvash and A. C. Parker, “A memristor SPICE model for designing memristor circuits,” Aug. 2010, pp. 989 –992.
[23] T. Chang, S. H. Jo, K. H. Kim, P. Sheridan, S. Gaba, W. Lu, “Synaptic behaviors and modeling of a metal oxide memristor device,” Applied Physics A, 102 pp. 857-863, (2011).
[24] H. Abdalla, and M. D. Pickett, “SPICE Modeling of Memristors,” *ISCAS* (2011).
[25] S. Shin, K. Kim, and S.-M. Kang, "Compact Models for Memristors Based on Charge-Flux Constitutive Relationships," *IEEE Transactions on Computer-Aided Design of Integrated Circuits and Systems*, vol. 29, no. 4, pp. 590-598, April 2010.
[26] J. G. Simmons, “Generalized Formula for the Electric Tunnel Effect between Similar Electrodes Separated by a Thin Insulating Film,” *Journal of Applied Physics*, 34(6), (1963).
[27] M. D. Pickett, D. B. Strukov, J. L. Borghetti, J. J. Yang, G. S. Snider, D. R. Stewart, and R. S. Williams, “Switching dynamics in titanium dioxide memristive devices,” *Journal of Applied Physics* 106(074508), (2009).